
Exigences

• Connaissances générales de 
l’étape datastep
• Énoncés set et merge

• « Program data vector » (PDV)

• SAS 9.2 ou plus récent

• Concept d’objet hash dans SAS

• Création d’un objet hash

• Utilisation de table SAS comme source

• Lecture de tables hash

• Écrire dans une table hash

• Gestion des doublons

• Itérateur hash

• Conclusion

• Références



Info sur les exemples

• Source: sashelp.cars

Source pour les 
exemples

sashelp.cars
• 428 observations
• 425 uniques selon make/model



Concept d’objet hash dans SAS

• L’objet hash fait parti d’une famille de 
composantes de type objet

• Méthode de stockage de donnée en mémoire 
pour référence future

• Permet la création et utilisation de vecteurs 
(arrays) multidimensionnels



Création d’un objet hash

1. Déclarer l’objet hash

declare hash <nom objet> ;

2. Instancier l’objet hash

<nom objet> = _new_ hash() ;

Ou encore, en une commande:

declare hash <nom objet>() ;



Création d’objet - la structure

• Une nouvelle table hash ne contient ni donnée, ni attributs

• Des méthodes hash (fonctions) servant à créer la structure
• <nom objet>.defineKey definie les clées d’index (jointure)

• <nom objet>.defineData definie les champs données de la table

• SAS ne fait rien avec ces énoncés tant que vous n’avez pas 
mentionné que la définition est terminée
• <nom objet>.definedone()

• Ces méthodes produisent un code de retour (rarement 
utilisé).  0 est un succès et autre valeur une erreur.



Utilisation de table SAS comme source

Une table SAS peut être utilisée comme source ou cible pour une table hash

• Source

Lorsqu’on instancie l’objet, on peut préciser la source

Declare hash <nom objet>(dataset:"<nom table SAS>")

• Cible

La méthode “output” sauvegarde la version actuelle de la table hash

<nom objet>.output(dataset:"<nom table SAS>")

• Vider une table

La méthode “clear” vide une table hash

<nom objet>.clear()



Cas 1: Charger une table SAS en mémoire

Cadre:

• Table SAS à charger en mémoire: sashelp.cars

• Variables clées: “make” et “model”

• Variables données: “MSRP” et “horsepower”

• Nom de la table hash à créer: ht_cars_ref

data _null_ ;

if 0 then set sashelp.cars(keep=make model msrp horsepower) ;

declare hash ht_cars_ref(dataset:"sashelp.cars") ;

ht_cars_ref.defineKey("make","model") ;

ht_cars_ref.defineData("msrp","horsepower") ;

ht_cars_ref.definedone() ;

run ;



Lecture de tables hash

• La façon la plus commune est la méthode “find”

• Find vérifie si une entrée existe pour la clé 
présente et retourne la/les valeur(s) le cas 
échéant

• La méthode “check” vérifie si une entrée existe 
pour la clé présente mais ne retourne pas 
d’information



Cas 2: Fusion de donnés non-triées

Cadre:

• On souhaite fusionner 2 tables sur les clés “make” et “model”
1. La table voitures_en_circulation contient 4B d’observations et 5000 variables non-triées

2. La table sashelp.cars (de notre cas 1) contient 428 lignes et 16 variables

data result_table ;

set voitures_en_circulation ;

/* Hash table definition */

if _n_ = 1 then do ;

if 0 then set sashelp.cars(keep=msrp horsepower) ;

declare hash ht_cars_ref(dataset:"sashelp.cars") ;

ht_cars_ref.defineKey("make","model") ;

ht_cars_ref.defineData("msrp","horsepower") ;

ht_cars_ref.definedone() ;

end ;

if ht_table_a.find() NE 0 then call missing(msrp,horsepower) ;

run ;



Qui veut devenir un maitre des tables hash ?

Question 1

Qu’arrive-t’il si la méthode find ne trouve pas de correspondance dans la table hash 
et que l’on n’effectue pas un call missing des variables?

Un avertissement est généré

Rien!

Une erreur est générée

Quelque chose



Écrire dans une table hash

• En instanciant une table hash, on peut charger une 
table SAS entière mais que fait-on si on souhaite 
ajouter des donnés après?

• La méthode “add” ajoute une nouvelle entrée…

… créant possiblement un doublon.

• La méthode “replace” ajoute une nouvelle entrée 
ou remplace l’existante (sans créer de doublons)

• La méthode “ref” créé une nouvelle entrée mais ne 
modifiera pas une entrée existante



Cas 3: Agrégation dans l’étape data

Cadre:

• On souhaite fusionner 2 tables sur les clés “make” et “model”
1. La table voitures_en_circulation contient 4B d’observations et 5000 variables 

non-triées

2. La table sashelp.cars contient 428 lignes et 16 variables

3. On veut créer une seconde table qui contient:
• Une ligne par valeur non-nulle de « make » se trouvant dans les 2 tables

• Un Compte de la fréquence de cette variable (nombre de lignes dans 
voitures_en_circulation)

• La somme des valeurs de la variable « msrp » (variable de la table sashelp.cars)



Case 3: Agrégation dans l’étape data
data result_table(drop=count_make total_make_msrp) ;

set voitures_en_circulation end=eof;

/* Hash table definition */

if _n_ = 1 then do ;

<definition ht_cars_ref (voir cas 2)>

declare hash ht_sum_table() ;

ht_sum_table.defineKey('make') ;

ht_sum_table.defineData('make','count_make','total_make_msrp') ;

ht_sum_table.definedone() ;

end ;

if ht_cars_ref.find() NE 0 then call missing(msrp,horsepower) ;

else do ;

if ht_sum_table.find() NE 0 then do ;

count_make = 0 ;

total_make_msrp = 0 ;

end ;

count_make = count_make + 1 ;

total_make_msrp = total_make_msrp + msrp ;

ht_sum_table.replace() ;

end ;

If eof then ht_sum_table.output(dataset:'work.summary_table') ;

run ;



Gestion des doublons

• Pour permettre les doublons dans une 
table hash, il faut préciser l’option 
“multidata:'Y'” lors de l’instanciation

• Sans cette option, seule la dernière 
occurrence d’une clé sera conserve dans 
la table hash

• La méthode “has_next” valide s’il y a au 
moins une autre entrée dans la table hash 
pour l’index courant

• La méthode “find_next” permet de lire 
l’entré suivante de la table hash



Gestion des doublons
• Pour naviguer à travers ces doublons, on 

peut suivre l’algorithme suivant:

find()

Has_next(result: r) r NE 0

match

fin

traitements

find_next()



Doublons dans leur habitat 
naturel

Besoins typiques pour des doublons :
• Relation un à plusieurs attendue

• Si on souhaite produire plusieurs “output” par 
ligne

• Relation un à plusieurs non-voulue
• Peut être utiliser pour créer un rapport 

d’erreur
• Permet de choisir la valeur la plus petite, 

grande, Moyenne, …



Outils aditionnels

Méthodes

• remove: retire la clée courante de la table hash (doublons 
inclus)

• delete: détruit l’objet hash (l’instance)

• equals: Compare un objet hash avec un autre

Attributs

• num_items: Nombre d’éléments de l’objet hash

• item_size: La taille de l’objet en byte (pour la clé 
seulement)



Itérateur Hash (l’autre objet hash)
• Table hash sans le concept des clés
• Mode de navigation pas à pas pour une table hash
• À chaque « pas », les valeurs sont retournées

Avantages Inconvénients

• Permet de parcourir l’objet 
sans index

• Approche moins performante
(vs utilisation index)

• Syntaxe peu intuitive



Itérateur Hash - la base

• Déclaration

declare hiter <nom objet>('<nom table hash>' );

• Méthodes pour la navigation

• Lire le premier élément de la table: “first”

• Lire le dernier élément de la table: “last”

• Lire l’élément suivant de la table: “next”

• Lire l’élément precedent de la table: “prev”



Cas 4: Agrégation avec match “flou”

Cadre:

• On souhaite fusionner 2 tables sur les clés “make” et “model”
1. La table voitures_en_circulation contient 4B d’observations et 5000 variables non-triées

2. La table sashelp.cars contient 428 lignes et 16 variables

3. On veut créer une seconde table qui contient:

• Une ligne par valeur non-nulle de make se trouvant dans les 2 tables

• Un Compte de la fréquence de cette variable (nombre de lignes dans voitures_en_circulation)

• La somme des valeurs de la variable msrp (variable de la table sashelp.cars)

4. Si « make » est manquant on permet une jointure sur le premier mot de la 
variable « model » et la puissance de la voiture



Cas 4: Agrégation avec match “flou”

• Exemple de match flou



Case 4: Agrégation avec match “flou”
data result_table(rename=(make_src = make model_src = model)) ;

set voitures_en_circulation end=eof;

drop make model ;

/* Hash table definition */

if _n_ = 1 then do ;

if 0 then set sashelp.cars(keep=msrp horsepower) ;

declare hash ht_cars_ref(dataset:"sashelp.cars") ;

declare hiter iter('ht_cars_ref');

ht_cars_ref.defineKey("make","model") ;

ht_cars_ref.defineData("make","model","msrp","horsepower") ;

ht_cars_ref.definedone() ;

declare hash ht_sum_table() ;

ht_sum_table.defineKey('make') ;

ht_sum_table.defineData('make','count_make','total_make_msrp') ;

ht_sum_table.definedone() ;

end ;

make_src = make  ;

model_src = model ;

Suite



Case 4: Agrégation avec match “flou” - suite
if ht_cars_ref.find() NE 0 then do ;

model_prfx = scan(model_src,1,' ') ;

match = 0 ;

rc = iter.first(); /* Aller chercher la première ligne de notre table hash */

do while (rc = 0);

model_prfx_ref = scan(model,1,' ') ;

if model_prfx_ref EQ model_prfx and puissance = horsepower then do ;

match = 1;

rc = 1 ;

end ;

else rc = iter.next();

end;

if match ne 1 then do ;

call missing(msrp,horsepower) ;  /* Case 2 code */

output ;

delete ;  /* force le passage à l’itération suivante du datastep */

end ;

end ;

Suite



Case 4: Agrégation avec match “flou” - suite et fin
/* Mise à jour donnés agrégées */

if ht_sum_table.find() NE 0 then do ;

count_make = 0 ;

total_make_msrp = 0 ;

end ;

count_make = count_make + 1 ;

total_make_msrp = total_make_msrp + msrp ;

ht_sum_table.replace() ;

output ;

If eof then ht_sum_table.output(dataset:'work.summary_table') ;

run ;

Puisque l’on a un énoncé “output” plus haut, on 
perd la sortie implicite de fin de datastep.
Il faut donc mettre un “output” explicite.



Qui veut devenir un maitre des tables hash ?

Question 2

Quand devriez-vous considérer l’utilisation de tables hash pour résoudre un 
problème SAS?

Les lundis et mercredis seulement

À 16h00 un vendredi avant 3 semaines de vacances

Pour gagner de la performance ou simplifier un code

Tout le temps, les tables hash sont vraiment cool!



Quand utiliser les objets hash

• Diminuer les “I/O”
• Joindre une table gigantesque avec une ou plusieurs petites tables sur 

différentes clés

• Simplifier le code (moins d’étapes)
• Création d’agrégations simple via une étape data existante (selon le même 

“group by” ou non)

• Éviter les limitations
• Utiliser comme vecteur pour éviter des limitations avec l’énoncé array



References

• Using DATA Step Component Objects:

http://support.sas.com/documentation/cdl/en/lrcon/65287/HTML/default/viewer.
htm#n0f1o4z0kjttssn15qnysjq1mzf3.htm

• SAS object tips sheet (cheat sheet):

https://support.sas.com/rnd/base/datastep/dot/hash-tip-sheet.pdf

http://support.sas.com/documentation/cdl/en/lrcon/65287/HTML/default/viewer.htm#n0f1o4z0kjttssn15qnysjq1mzf3.htm
https://support.sas.com/rnd/base/datastep/dot/hash-tip-sheet.pdf



